

# Numeri complessi

## - Note

### - Autore

Claudio Marsan  
Liceo Cantonale di Mendrisio  
Via Agostino Maspoli  
CH-6850 Mendrisio (Switzerland)  
e-mail: claudio.marsan@liceomendrisio.ch

### - Versione

Versione 2.0, 16 marzo 2003  
Maple V Release 6.02 for Windows 2000

```
> restart: with(plots):  
Warning, the name changecoords has been redefined
```

L'unità immaginaria è I (maiuscola!):

```
> sqrt(-1);  
  
I  
  
> I^2;  
  
-1  
  
> seq(I^k, k=0..12);  
  
1, I, -1, -I, 1, I, -1, -I, 1, I, -1, -I, 1
```

Operazioni elementari con i numeri complessi:

```
> z1 := a + b*I; z2 := c + d*I;  
  
z1 := a + I b  
z2 := c + I d  
  
> z1 + z2;  
z1 - z2;  
z1 * z2;  
z1 / z2;  
  
a + I b + c + I d  
a + I b - c - I d  
(a + I b)(c + I d)  
a + I b  
c + I d
```

Per avere i risultati nella forma rettangolare bisogna usare il comando evalc (evaluate complex):

```
> evalc(z1 + z2);  
evalc(z1 - z2);  
evalc(z1 * z2);
```

```
evalc(z1 / z2);
```

$$\begin{aligned} & a + c + I(b + d) \\ & a - c + I(b - d) \\ & ac - bd + I(ad + bc) \end{aligned}$$

$$\frac{ac}{c^2 + d^2} + \frac{bd}{c^2 + d^2} + I \left( \frac{bc}{c^2 + d^2} - \frac{ad}{c^2 + d^2} \right)$$

```
> `(3 + 2*I) * (5 - 4*I) + 1/(I^3 - 2*I^2 + 1)` = (3 + 2*I) * (5 - 4*I) + 1/(I^3 - 2*I^2 + 1);
```

$$(3 + 2I) * (5 - 4I) + 1/(I^3 - 2I^2 + 1) = \frac{233}{10} - \frac{19}{10}I$$

Parte reale e parte immaginaria di un numero complesso z; valore assoluto di z; coniugato complesso di z:

```
> z := 5 - 3*I;
`Re(z)` = Re(z);
`Im(z)` = Im(z);
`|z|` = abs(z);
`conj(z)` = conjugate(z);
```

$$z := 5 - 3I$$

$$\operatorname{Re}(z) = 5$$

$$\operatorname{Im}(z) = -3$$

$$|z| = \sqrt{34}$$

$$\operatorname{conj}(z) = 5 + 3I$$

Equazioni con i numeri complessi:

```
> z := 'z':
solve(z^3 = 2 + 3*I, z);
```

$$(2 + 3I)^{(1/3)}, -\frac{1}{2}(2 + 3I)^{(1/3)} + \frac{1}{2}I\sqrt{3}(2 + 3I)^{(1/3)},$$

$$-\frac{1}{2}(2 + 3I)^{(1/3)} - \frac{1}{2}I\sqrt{3}(2 + 3I)^{(1/3)}$$

```
> evalf(%);
```

$$1.451856618 + .4934035341I, -1.153228304 + 1.010642947I, -2.986283140 - 1.504046481I$$

La funzione `nth_root(z, n)` calcola le  $n$  radici  $n$ -esime di un numero complesso  $z$  e disegna il relativo poligono nel piano di Gauss.

```
> nth_root := proc(z::complex, n::posint)
local i, S, r, phi;
r := (abs(z))^(1/n);
phi := argument(z);
S := seq([r*cos((phi+2*i*Pi)/n), r*sin((phi+2*i*Pi)/n)], i =
0..n-1);
```

```

    print(S);
    plot([S, S[1]], x=-r..r, y=-r..r, style=LINE,
scaling=CONSTRAINED);
end;


nth_root := proc(z::complex, n::posint)
local i, S, r, phi;
    r := abs(z)^(1 / n);
    phi := argument(z);
    S := seq([r*cos((phi + 2*i*pi) / n), r*sin((phi + 2*i*pi) / n)], i = 0 .. n - 1);
    print(S);
    plot([S, S[1]], x = -r .. r, y = -r .. r, style = LINE, scaling = CONSTRAINED)
end proc

```

Qualche esempio:


```
> nth_root(1, 3);
```

$[1, 0], \left[\frac{-1}{2}, \frac{1}{2}\sqrt{3}\right], \left[\frac{-1}{2}, -\frac{1}{2}\sqrt{3}\right]$



```
> nth_root(1, 4);
```


$[1, 0], [0, 1], [-1, 0], [0, -1]$



```
> nth_root(-64, 6);
```

$$\left[ \frac{1}{2} 64^{(1/6)} \sqrt{3}, \frac{1}{2} 64^{(1/6)} \right], [0, 64^{(1/6)}], \left[ -\frac{1}{2} 64^{(1/6)} \sqrt{3}, \frac{1}{2} 64^{(1/6)} \right],$$


$$\left[ -\frac{1}{2} 64^{(1/6)} \sqrt{3}, -\frac{1}{2} 64^{(1/6)} \right], [0, -64^{(1/6)}], \left[ \frac{1}{2} 64^{(1/6)} \sqrt{3}, -\frac{1}{2} 64^{(1/6)} \right]$$



```
> nth_root(-5, 5);
```

$$\left[ 5^{(1/5)} \cos\left(\frac{1}{5} \pi\right), 5^{(1/5)} \sin\left(\frac{1}{5} \pi\right) \right], \left[ -5^{(1/5)} \cos\left(\frac{2}{5} \pi\right), 5^{(1/5)} \sin\left(\frac{2}{5} \pi\right) \right], \left[ -5^{(1/5)}, 0 \right],$$

$$\left[ -5^{(1/5)} \cos\left(\frac{2}{5} \pi\right), -5^{(1/5)} \sin\left(\frac{2}{5} \pi\right) \right], \left[ 5^{(1/5)} \cos\left(\frac{1}{5} \pi\right), -5^{(1/5)} \sin\left(\frac{1}{5} \pi\right) \right]$$



> **nth\_root(1, 17);**

$$\left[ 1, 0 \right], \left[ \cos\left(\frac{2}{17} \pi\right), \sin\left(\frac{2}{17} \pi\right) \right], \left[ \cos\left(\frac{4}{17} \pi\right), \sin\left(\frac{4}{17} \pi\right) \right], \left[ \cos\left(\frac{6}{17} \pi\right), \sin\left(\frac{6}{17} \pi\right) \right],$$


$$\left[ \cos\left(\frac{8}{17} \pi\right), \sin\left(\frac{8}{17} \pi\right) \right], \left[ -\cos\left(\frac{7}{17} \pi\right), \sin\left(\frac{7}{17} \pi\right) \right], \left[ -\cos\left(\frac{5}{17} \pi\right), \sin\left(\frac{5}{17} \pi\right) \right],$$

$$\left[ -\cos\left(\frac{3}{17} \pi\right), \sin\left(\frac{3}{17} \pi\right) \right], \left[ -\cos\left(\frac{1}{17} \pi\right), \sin\left(\frac{1}{17} \pi\right) \right], \left[ -\cos\left(\frac{1}{17} \pi\right), -\sin\left(\frac{1}{17} \pi\right) \right],$$

$$\left[ -\cos\left(\frac{3}{17} \pi\right), -\sin\left(\frac{3}{17} \pi\right) \right], \left[ -\cos\left(\frac{5}{17} \pi\right), -\sin\left(\frac{5}{17} \pi\right) \right], \left[ -\cos\left(\frac{7}{17} \pi\right), -\sin\left(\frac{7}{17} \pi\right) \right],$$

$$\left[ \cos\left(\frac{8}{17} \pi\right), -\sin\left(\frac{8}{17} \pi\right) \right], \left[ \cos\left(\frac{6}{17} \pi\right), -\sin\left(\frac{6}{17} \pi\right) \right], \left[ \cos\left(\frac{4}{17} \pi\right), -\sin\left(\frac{4}{17} \pi\right) \right],$$

$$\left[ \cos\left(\frac{2}{17} \pi\right), -\sin\left(\frac{2}{17} \pi\right) \right]$$



Conversione di un numero complesso dalla forma rettangolare a quella polare:

> `convert(2 + 2*I, polar);`

$$\text{polar}\left(2\sqrt{2}, \frac{1}{4}\pi\right)$$

> `convert(-3*I, polar);`

$$\text{polar}\left(3, -\frac{1}{2}\pi\right)$$

Conversione di un numero complesso dalla forma polare a quella rettangolare:

> `evalc(polar(2*sqrt(2), Pi/4));`

$$2 + 2I$$

> `evalc(polar(3, -1/2*Pi));`

$$-3I$$

Conversione di un numero complesso dalla forma esponenziale a quella rettangolare:

> `evalc(exp(a+b*I));`

$$e^a \cos(b) + I e^a \sin(b)$$

> `evalc(exp(Pi*I));`

$$-1$$

Conversione di un numero complesso dalla forma esponenziale a quella polare:

> `convert(exp(Pi*I), polar);`

$$\text{polar}(1, \pi)$$

> `polar(z);`

polar(|z|, argument(z))

Una possibile definizione della funzione cis (da rettangolare a trigonometrica, con angoli in gradi):

```
> cis := z -> abs(z) * ('cos'(argument(z)*180/Pi*DEG) +  
I*'sin'(argument(z)*180/Pi*DEG));
```

$$cis := z \rightarrow |z| \left( \cos \left( 180 \frac{\text{argument}(z) \text{ DEG}}{\pi} \right) + I \sin \left( 180 \frac{\text{argument}(z) \text{ DEG}}{\pi} \right) \right)$$

```
> cis(5 + 5*I);
```

$$5\sqrt{2} (\cos(45 \text{ DEG}) + I \sin(45 \text{ DEG}))$$

```
> cis(sqrt(3) - 3*I);
```

$$2\sqrt{3} (\cos(-60 \text{ DEG}) + I \sin(-60 \text{ DEG}))$$

I più importanti numeri sono legati da questa straordinaria relazione, nota come **formula di Eulero**

:

```
> 'exp(Pi * I)' + 1 = 0;
```

$$e^{(I\pi)} + 1 = 0$$

```
> eval(%);
```

$$0 = 0$$